
Introduction

➤ There was an urgent need for a common framework.

➤ CoFI aims at establishing a wide consensus.

➤ The focus of CoFI is on algebraic techniques.

➤ CoFI has already achieved its main aims.

➤ CoFI is an open, voluntary initiative.

➤ CoFI has received funding as an ESPRIT Working Group,

and is sponsored by IFIP WG 1.3.

➤ New participants are welcome!

➤ CASL has been designed as a general-purpose algebraic specification language,

subsuming many existing languages.

➤ CASL is at the center of a family of languages.

The CASL Family of Languages

➤ CASL itself has several major parts.

Underlying Concepts

➤ CASL is based on standard concepts of algebraic specification.

➤ Basic specifications.

• A basic specification declares symbols, and gives axioms and constraints.

• The semantics of a basic specification is a signature and a class of models.

• CASL specifications may declare sorts, subsorts, operations, and predicates.

• Subsorts declarations are interpreted as embeddings.

• Operations may be declared as total or partial.

• Predicates are different from boolean-valued operations.

• Operation symbols and predicate symbols may be overloaded.

• Axioms are formulas of first-order logic.

• Sort generation constraints eliminate ‘junk’ from specific carrier sets.

➤ Structured specifications.

• The semantics of a structured specification is simply a signature and a class of

models.

• A translation merely renames symbols.

• Hiding symbols removes parts of models.

• Union of specifications identifies common symbols.

• Extension of specifications identifies common symbols too.

• Free specifications restrict models to being free, with initiality as a special case.

• Generic specifications have parameters, and have to be instantiated when

referenced.

➤ Architectural specifications and Libraries.

• The semantics of an architectural specification reflects its modular structure.

• Architectural specifications involve the notions of persistent function and

conservative extension.

• The semantics of a library of specifications is a mapping from the names of the

specifications to their semantics.

Foundations

➤ A complete presentation of CASL is in the Reference Manual

• CASL has a definitive summary.

• CASL has a complete formal definition.

• Abstract and concrete syntax of CASL are defined formally.

• CASL has a complete formal semantics.

• CASL specifications denote classes of models.

• The semantics is largely institution-independent.

• The semantics is the ultimate reference for the meanings of all CASL constructs.

• Proof systems for various layers of CASL are provided.

• A formal refinement concept for CASL specifications is introduced.

• The foundations of our CASL are rock-solid!

Getting Started

➤ Simple specifications may be written in CASL essentially as in many other algebraic

specification languages.

➤ CASL provides also useful abbreviations.

➤ CASL allows loose, generated and free specifications.

Loose Specifications

➤ CASL syntax for declarations and axioms involves familiar notation, and is mostly

self-explanatory.

spec Strict Partial Order =

%% Let’s start with a simple example !

sort Elem

pred < : Elem × Elem %% pred abbreviates predicate

∀x , y , z : Elem

• ¬(x < x ) %(strict)%

• x < y ⇒ ¬(y < x ) %(asymmetric)%

• x < y ∧ y < z ⇒ x < z %(transitive)%

%{ Note that there may exist x, y such that

neither x < y nor y < x. }%

end

➤ Specifications can easily be extended by new declarations and axioms.

spec Total Order =

Strict Partial Order

then ∀x , y : Elem • x < y ∨ y < x ∨ x = y %(total)%

end

➤ In simple cases, an operation (or a predicate) symbol may be declared and its

intended interpretation defined at the same time.

spec Total Order With MinMax =

Total Order

then ops min(x , y : Elem) : Elem = x when x < y else y ;

max (x , y : Elem) : Elem = y when min(x , y) = x else x

end

spec Variant Of Total Order With MinMax =

Total Order

then vars x , y : Elem

op min : Elem × Elem → Elem

• x < y ⇒ min(x , y) = x

• ¬(x < y) ⇒ min(x , y) = y

op max : Elem × Elem → Elem

• x < y ⇒ max (x , y) = y

• ¬(x < y) ⇒ max (x , y) = x

end

➤ Symbols may be conveniently displayed as usual mathematical symbols by means of

%display annotations.

%display <= %LATEX ≤

spec Partial Order =

Strict Partial Order

then pred ≤ (x , y : Elem) ⇔ (x < y ∨ x = y)

end

➤ The %implies annotation is used to indicate that some axioms are supposedly

redundant, being consequences of others.

spec Partial Order 1 =

Partial Order

then %implies

∀x , y , z : Elem • x ≤ y ∧ y ≤ z ⇒ x ≤ z %(transitive)%

end

spec Implies Does Not Hold =

Partial Order

then %implies

∀x , y : Elem • x < y ∨ y < x ∨ x = y %(total)%

end

➤ Attributes may be used to abbreviate axioms for associativity, commutativity,

idempotence, and unit properties.

spec Monoid =

sort Monoid

ops 1 : Monoid ;

∗ : Monoid × Monoid → Monoid , assoc, unit 1

end

➤ Genericity of specifications can be made explicit using parameters.

spec Generic Monoid [ sort Elem ] =

sort Monoid

ops inj : Elem → Monoid ;

1 : Monoid ;

∗ : Monoid × Monoid → Monoid , assoc, unit 1

∀x , y : Elem • inj (x ) = inj (y) ⇒ x = y

end

spec Non Generic Monoid =

sort Elem

then sort Monoid

ops inj : Elem → Monoid ;

1 : Monoid ;

∗ : Monoid × Monoid → Monoid , assoc, unit 1

∀x , y : Elem • inj (x ) = inj (y) ⇒ x = y

end

➤ References to generic specifications always instantiate the parameters.

spec Generic Commutative Monoid [ sort Elem ] =

Generic Monoid [ sort Elem ]

then ∀x , y : Monoid • x ∗ y = y ∗ x

end

spec Generic Commutative Monoid 1 [ sort Elem ] =

Generic Monoid [ sort Elem ]

then op ∗ : Monoid × Monoid → Monoid , comm

end

➤ Datatype declarations may be used to abbreviate declarations of sorts and

constructors.

spec Container [ sort Elem ] =

type Container ::= empty | insert(Elem; Container)

pred is in : Elem × Container

∀e, e ′ : Elem; C : Container

• ¬(e is in empty)

• e is in insert(e ′,C ) ⇔ (e = e ′ ∨ e is in C )

end

➤ Loose datatype declarations are appropriate when further constructors may be

added in extensions.

spec Marking Container [ sort Elem ] =

Container [ sort Elem ]

then type Container ::= mark insert(Elem; Container)

pred is marked in : Elem × Container

∀e, e ′ : Elem; C : Container

• e is in mark insert(e ′,C ) ⇔ (e = e ′ ∨ e is in C )

• ¬(e is marked in empty)

• e is marked in insert(e ′,C ) ⇔ e is marked in C

• e is marked in mark insert(e ′,C ) ⇔ (e = e ′ ∨ e is marked in C )

end

Generated Specifications

➤ Sorts may be specified as generated by their constructors.

spec Generated Container [ sort Elem ] =

generated type Container ::= empty | insert(Elem; Container)

pred is in : Elem × Container

∀e, e ′ : Elem; C : Container

• ¬(e is in empty)

• e is in insert(e ′,C ) ⇔ (e = e ′ ∨ e is in C )

end

➤ Generated specifications are in general loose.

spec Generated Container Merge [ sort Elem ] =

Generated Container [ sort Elem ]

then op merge : Container × Container → Container

∀e : Elem; C ,C ′ : Container

• e is in (C merge C ′) ⇔ (e is in C ∨ e is in C ′)

end

➤ Generated specifications need not be loose.

spec Generated Set [ sort Elem ] =

generated type Set ::= empty | insert(Elem; Set)

pred is in : Elem × Set

ops { }(e : Elem) : Set = insert(e, empty);

∪ : Set × Set → Set ;

remove : Elem × Set → Set

∀e, e′ : Elem; S ,S ′ : Set

• ¬(e is in empty)

• e is in insert(e′,S) ⇔ (e = e′ ∨ e is in S)

• S = S ′ ⇔ (∀x : Elem • x is in S ⇔ x is in S ′) %(equal sets)%

• e is in (S ∪ S ′) ⇔ (e is in S ∨ e is in S ′)

• e is in remove(e′,S) ⇔ (¬(e = e′) ∧ e is in S)

then %implies ∀e, e′ : Elem; S : Set

• insert(e, insert(e,S)) = insert(e,S)

• insert(e, insert(e′,S)) = insert(e′, insert(e,S))

generated type Set ::= empty | { }(Elem) | ∪ (Set ; Set)

op ∪ : Set × Set → Set , assoc, comm, idem, unit empty

end

➤ Generated types may need to be declared together.

sort Node

generated type Tree ::= mktree(Node; Forest)

generated type Forest ::= empty | add(Tree; Forest)

is both incorrect (linear visibility) and wrong (the corresponding semantics is not the

“expected” one). One must write instead:

sort Node

generated types Tree ::= mktree(Node; Forest);

Forest ::= empty | add(Tree; Forest)

Free Specifications

➤ Free specifications provide initial semantics and avoid the need for explicit negation.

spec Natural = free type Nat ::= 0 | suc(Nat)

➤ Free datatype declarations are particularly convenient for defining enumerated

datatypes.

spec Color =

free type RGB ::= Red | Green | Blue

free type CMYK ::= Cyan | Magenta | Yellow | Black

end

➤ Free specifications can also be used when the constructors are related by some

axioms.

spec Integer =

free { type Int ::= 0 | suc(Int) | pre(Int)

∀x : Int • suc(pre(x )) = x

• pre(suc(x )) = x }

end

➤ Predicates hold minimally in models of free specifications.

spec Natural Order =

Natural

then free { pred < : Nat × Nat

∀x , y : Nat

• 0 < suc(x )

• x < y ⇒ suc(x ) < suc(y) }

end

➤ Operations and predicates may be safely defined by induction on the constructors of

a free datatype declaration.

spec Natural Arithmetic =

Natural Order

then ops 1 : Nat = suc(0 );

+ : Nat × Nat → Nat , assoc, comm, unit 0 ;

∗ : Nat × Nat → Nat , assoc, comm, unit 1

∀x , y : Nat

• x + suc(y) = suc(x + y)

• x ∗ 0 = 0

• x ∗ suc(y) = (x ∗ y) + x

end

➤ More care may be needed when defining operations or predicates on free datatypes

when there are axioms relating the constructors.

spec Integer Arithmetic =

Integer

then ops 1 : Int = suc(0 );

+ : Int × Int → Int , assoc, comm, unit 0 ;

− : Int × Int → Int ;

∗ : Int × Int → Int , assoc, comm, unit 1

∀x , y : Int

• x + suc(y) = suc(x + y)

• x + pre(y) = pre(x + y)

• x − 0 = x

• x − suc(y) = pre(x − y)

• x − pre(y) = suc(x − y)

• x ∗ 0 = 0

• x ∗ suc(y) = (x ∗ y) + x

• x ∗ pre(y) = (x ∗ y) − x

end

spec Integer Arithmetic Order =

Integer Arithmetic

then preds ≤ , ≥ , < , > : Int × Int

∀x , y : Int

• 0 ≤ 0

• ¬(0 ≤ pre(0 ))

• 0 ≤ x ⇒ 0 ≤ suc(x )

• ¬(0 ≤ x ) ⇒ ¬(0 ≤ pre(x ))

• suc(x ) ≤ y ⇔ x ≤ pre(y)

• pre(x ) ≤ y ⇔ x ≤ suc(y)

• x ≥ y ⇔ y ≤ x

• x < y ⇔ (x ≤ y ∧ ¬(x = y))

• x > y ⇔ y < x

end

➤ Generic specifications often involve free extensions of (loose) parameters.

spec List [ sort Elem ] = free type List ::= empty | cons(Elem; List)

spec Set [ sort Elem ] =

free { type Set ::= empty | insert(Elem; Set)

pred is in : Elem × Set

∀e, e ′ : Elem; S : Set

• insert(e, insert(e,S )) = insert(e,S )

• insert(e, insert(e ′,S )) = insert(e ′, insert(e,S ))

• ¬(e is in empty)

• e is in insert(e,S )

• e is in insert(e ′,S ) if e is in S }

end

spec Transitive Closure [ sort Elem pred R : Elem × Elem ] =

free { pred R+ : Elem × Elem

∀x , y , z : Elem

• x R y ⇒ x R+y

• x R+y ∧ y R+z ⇒ x R+z }

➤ Loose extensions of free specifications can avoid overspecification.

spec Natural With Bound =

Natural Arithmetic

then op max size : Nat

• 0 < max size

end

spec Set Choose [ sort Elem ] =

Set [ sort Elem ]

then op choose : Set → Elem

∀S : Set • ¬(S = empty) ⇒ choose(S ) is in S

end

➤ Datatypes with observer operations or predicates can be specified as generated

instead of free.

spec Set Generated [ sort Elem ] =

generated type Set ::= empty | insert(Elem; Set)

pred is in : Elem × Set

∀e, e ′ : Elem; S ,S ′ : Set

• ¬(e is in empty)

• e is in insert(e ′,S ) ⇔ (e = e ′ ∨ e is in S )

• S = S ′ ⇔ (∀x : Elem • x is in S ⇔ x is in S ′)

end

➤ The %def annotation is useful to indicate that some operations or predicates are

uniquely defined.

spec Set Union [ sort Elem ] =

Set [ sort Elem ]

then %def

ops ∪ : Set × Set → Set , assoc, comm, idem, unit empty ;

remove : Elem × Set → Set

∀e, e ′ : Elem; S ,S ′ : Set

• S ∪ insert(e ′,S ′) = insert(e ′,S ∪ S ′)

• remove(e, empty) = empty

• remove(e, insert(e,S )) = remove(e,S )

• remove(e, insert(e ′,S )) = insert(e ′, remove(e,S )) if ¬(e = e ′)

end

➤ Operations can be defined by axioms involving observer operations, instead of

inductively on constructors.

spec Set Union 1 [ sort Elem ] =

Set Generated [ sort Elem ]

then %def

ops ∪ : Set × Set → Set , assoc, comm, idem, unit empty ;

remove : Elem × Set → Set

∀e, e ′ : Elem; S ,S ′ : Set

• e is in (S ∪ S ′) ⇔ (e is in S ∨ e is in S ′)

• e is in remove(e ′,S ) ⇔ (¬(e = e ′) ∧ e is in S )

end

➤ Sorts declared in free specifications are not necessarily generated by their

constructors.

spec UnNatural =

free { type UnNat ::= 0 | suc(UnNat)

op + : UnNat × UnNat → UnNat ,

assoc, comm, unit 0

∀x , y : UnNat • x + suc(y) = suc(x + y)

∀x : UnNat • ∃y : UnNat • x + y = 0 }

end

Partial Functions

➤ Partial functions arise naturally.

➤ Partial functions are declared differently from total functions.

spec Set Partial Choose [ sort Elem ] =

Generated Set [ sort Elem ]

then op choose : Set →? Elem

end

➤ Terms containing partial functions may be undefined, i.e., they may fail to denote

any value.

E.g., the (value of the) term choose(empty) may be undefined.

➤ Functions, even total ones, propagate undefinedness.

If the term choose(S ) is undefined for some value of S ,

then the term insert(choose(S ),S ′) is undefined as well for this value of S ,

although insert is a total function.

➤ Predicates do not hold on undefined arguments.

If the term choose(S ) is undefined,

then the atomic formula choose(S ) is in S does not hold.

➤ Equations hold when both terms are undefined.

The ordinary equation:

insert(choose(S ), insert(choose(S ), empty)) = insert(choose(S ), empty)

holds also when the term choose(S ) is undefined.

➤ Special care is needed in specifications involving partial functions.

• Asserting choose(S ) is in S as an axiom

implies that choose(S ) is defined, for any S .

• Asserting remove(choose(S ), insert(choose(S ), empty)) = empty

as an axiom implies that choose(S ) is defined for any S ,

since the term empty is always defined.

• Asserting insert(choose(S ),S ) = S as an axiom

implies that choose(S ) is defined for any S ,

since a variable always denotes a defined value.

➤ The definedness of a term can be checked or asserted.

spec Set Partial Choose 1 [ sort Elem ] =

Set Partial Choose [ sort Elem ]

then • ¬ def choose(empty)

∀S : Set • def choose(S ) ⇒ choose(S ) is in S

end

We know that choose is undefined when applied to empty ,

but we don’t know exactly when choose(S ) is defined.

(It may be undefined on other values than empty .)

If we would have specified choose by:

∀S : Set • ¬(S = empty) ⇒ choose(S ) is in S

then we could conclude that choose(S ) is defined when S is not equal to empty ,

but nothing about the undefinedness of choose(empty).

➤ The domains of definition of partial functions can be specified exactly.

spec Set Partial Choose 2 [ sort Elem ] =

Set Partial Choose [ sort Elem ]

then ∀S : Set • def choose(S ) ⇔ ¬(S = empty)

∀S : Set • def choose(S ) ⇒ choose(S ) is in S

end

➤ Loosely specified domains of definition may be useful.

spec Natural With Bound And Addition =

Natural With Bound

then op +? : Nat × Nat →? Nat

∀x , y : Nat

• def (x+?y) if x + y < max size

%{ x + y < max size implies both

x < max size and y < max size }%

• def (x+?y) ⇒ x+?y = x + y

end

➤ Domains of definition can be specified more or less explicitly.

spec Set Partial Choose 3 [ sort Elem ] =

Set Partial Choose [ sort Elem ]

then • ¬ def choose(empty)

∀S : Set • ¬(S = empty) ⇒ choose(S ) is in S

end

We can conclude after some reasoning that:

def choose(S ) ⇔ ¬(S = empty)

but this is not so prominent.

spec Natural Partial Pre =

Natural Arithmetic

then op pre : Nat →? Nat

• ¬ def pre(0 )

∀x : Nat • pre(suc(x )) = x

end

is explicit enough.

spec Natural Partial Subtraction 1 =

Natural Partial Pre

then op − : Nat × Nat →? Nat

∀x , y : Nat

• x − 0 = x

• x − suc(y) = pre(x − y)

end

is correct, but clearly not explicit enough, and better specified as follows:

spec Natural Partial Subtraction =

Natural Partial Pre

then op − : Nat × Nat →? Nat

∀x , y : Nat

• def (x − y) ⇔ (y < x ∨ y = x )

• x − 0 = x

• x − suc(y) = pre(x − y)

end

➤ Partial functions are minimally defined by default in free specifications.

spec List Selectors 1 [ sort Elem ] =

List [ sort Elem ]

then free { ops head : List →? Elem;

tail : List →? List

∀e : Elem; L : List

• head(cons(e,L)) = e

• tail(cons(e,L)) = L }

end

spec List Selectors 2 [ sort Elem ] =

List [ sort Elem ]

then ops head : List →? Elem;

tail : List →? List

∀e : Elem; L : List

• ¬ def head(empty)

• ¬ def tail(empty)

• head(cons(e,L)) = e

• tail(cons(e,L)) = L

end

➤ Selectors can be specified concisely in datatype declarations, and are usually partial.

spec List Selectors [ sort Elem ] =

free type List ::= empty | cons(head :? Elem; tail :? List)

spec Natural Suc Pre = free type Nat ::= 0 | suc(pre :? Nat)

➤ Selectors are usually total when there is only one constructor.

spec Pair 1 [ sorts Elem1 , Elem2 ] =

free type Pair ::= pair(first : Elem1 ; second : Elem2 )

➤ Constructors may be partial.

spec Part Container [ sort Elem ] =

generated type

P Container ::= empty | insert(Elem; P Container)?

pred addable : Elem × P Container

vars e, e ′ : Elem; C : P Container

• def insert(e,C ) ⇔ addable(e,C )

pred is in : Elem × P Container

• ¬(e is in empty)

• (e is in insert(e ′,C ) ⇔ (e = e ′ ∨ e is in C )) if addable(e ′,C )

end

➤ Existential equality requires the definedness of both terms as well as their equality.

spec Natural Partial Subtraction 2 =

Natural Partial Subtraction 1

then ∀x , y , z : Nat • y − x
e
= z − x ⇒ y = z

%{ y − x = z − x ⇒ y = z would be wrong,

def (y − x) ∧ def (z − x) ∧ y − x = z − x ⇒ y = z

is correct, but better abbreviated in the above axiom }%

end

Subsorting

➤ Subsorts and supersorts are often useful in CASL specifications.

➤ Subsort declarations directly express relationships between carrier sets.

spec Generic Monoid 1 [ sort Elem ] =

sorts Elem < Monoid

ops 1 : Monoid ;

∗ : Monoid × Monoid → Monoid , assoc, unit 1

end

➤ Operations declared on a sort are automatically inherited by its subsorts.

spec Vehicle =

Natural

then sorts Car , Bicycle < Vehicle

ops max speed : Vehicle → Nat ;

weight : Vehicle → Nat ;

engine capacity : Car → Nat

end

➤ Inheritance applies also for subsorts that are declared afterwards.

spec More Vehicle = Vehicle then sorts Boat < Vehicle

➤ Subsort membership can be checked or asserted.

spec Speed Regulation =

Vehicle

then ops speed limit : Vehicle → Nat ;

car speed limit , bike speed limit : Nat

∀v : Vehicle

• v ∈ Car ⇒ speed limit(v) = car speed limit

• v ∈ Bicycle ⇒ speed limit(v) = bike speed limit

end

➤ Datatype declarations can involve subsort declarations.

sorts Car , Bicycle, Boat

type Vehicle ::= sort Car | sort Bicycle | sort Boat

is equivalent to the declaration sorts Car , Bicycle, Boat < Vehicle,

and leaves the way open to further kinds of vehicles (e.g., planes).

sorts Car , Bicycle, Boat

generated type Vehicle ::= sort Car | sort Bicycle | sort Boat

prevents the definition of further subsorts, e.g., for planes.

sorts Car , Bicycle, Boat

free type Vehicle ::= sort Car | sort Bicycle | sort Boat

prevents the definition of further subsorts, and moreover the definition of a common

subsort of both Car and Boat (e.g., sorts Amphibious < Car ,Boat).

➤ Subsorts may also arise as classifications of previously specified values, and their

values can be explicitly defined.

spec Natural Subsorts =

Natural Arithmetic

then pred even : Nat

• even(0 )

• ¬ even(1 )

∀n : Nat • even(suc(suc(n))) ⇔ even(n)

sort Even = {x : Nat • even(x )}

sort Prime = {x : Nat • 1 < x ∧

∀y , z : Nat • x = y ∗ z ⇒ y = 1 ∨ z = 1}

end

spec Positive =

Natural Partial Pre

then sort Pos = {x : Nat • ¬(x = 0 )}

➤ It may be useful to redeclare previously defined operations, using the new subsorts

introduced.

spec Positive Arithmetic =

Positive

then ops 1 : Pos;

suc : Nat → Pos;

+ , ∗ : Pos × Pos → Pos;

+ : Pos × Nat → Pos;

+ : Nat × Pos → Pos

end

➤ A subsort may correspond to the definition domain of a partial function.

spec Positive Pre =

Positive Arithmetic

then op pre : Pos → Nat

➤ Using subsorts may avoid the need for partial functions.

spec Natural Positive Arithmetic =

free types Nat ::= 0 | sort Pos;

Pos ::= suc(pre : Nat)

ops 1 : Pos = suc(0 );

+ : Nat × Nat → Nat , assoc, comm, unit 0 ;

∗ : Nat × Nat → Nat , assoc, comm, unit 1 ;

+ , ∗ : Pos × Pos → Pos;

+ : Pos × Nat → Pos;

+ : Nat × Pos → Pos

∀x , y : Nat

• x + suc(y) = suc(x + y)

• x ∗ 0 = 0

• x ∗ suc(y) = x + (x ∗ y)

end

➤ Casting a term from a supersort to a subsort is explicit and the value of the cast

may be undefined.

Casting a term t to a sort s is written t as s,

and def (t as s) is equivalent to t ∈ s.

• pre( pre(suc(1 )) as Pos )

• def pre( pre(suc(1 )) as Pos )

• ¬ def (pre( pre(suc(1 )) as Pos ) as Pos)

➤ Supersorts may be useful when generalizing previously specified sorts.

spec Integer Arithmetic 1 =

Natural Positive Arithmetic

then free type Int ::= sort Nat | − (Pos)

ops + : Int × Int → Int , assoc, comm, unit 0 ;

− : Int × Int → Int ;

∗ : Int × Int → Int , assoc, comm, unit 1

∀x : Int ; n : Nat ; p, q : Pos

• suc(n) + (−1 ) = n

• suc(n) + (−suc(q)) = n + (−q)

• (−p) + (−q) = −(p + q)

• x − 0 = x

• x − p = x + (−p)

• x − (−q) = x + q

• 0 ∗ (−q) = 0

• p ∗ (−q) = −(p ∗ q)

• (−p) ∗ (−q) = p ∗ q

end

➤ Supersorts may also be used for extending the intended values by new values

representing errors or exceptions.

spec Set Error Choose [ sort Elem ] =

Generated Set [ sort Elem ]

then sorts Elem < ElemError

op choose : Set → ElemError

pred is in : ElemError × Set

∀S : Set • ¬(S = empty) ⇒ choose(S ) ∈ Elem ∧ choose(S ) is in S

end

spec Set Error Choose 1 [ sort Elem ] =

Generated Set [ sort Elem ]

then sorts Elem < ElemError

op choose : Set → ElemError

∀S : Set • ¬(S = empty) ⇒ (choose(S ) as Elem) is in S

end

Structuring Specifications

➤ Large and complex specifications are easily built out of simpler ones by means of

(a small number of) specification-building operations.

➤ Union and extension can be used to structure specifications.

spec List Set [ sort Elem ] =

List Selectors [ sort Elem ]

and Generated Set [ sort Elem ]

then op elements of : List → Set

∀e : Elem; L : List

• elements of empty = empty

• elements of cons(e,L) = {e} ∪ elements of L

end

➤ Specifications may combine parts with loose, generated, and free interpretations.

spec List Choose [ sort Elem ] =

List Selectors [ sort Elem ]

and Set Partial Choose 2 [ sort Elem ]

then ops elements of : List → Set ;

choose : List →? Elem

∀e : Elem; L : List

• elements of empty = empty

• elements of cons(e,L) = {e} ∪ elements of L

• def choose(L) ⇔ ¬(L = empty)

• choose(L) = choose(elements of L)

end

c© Michel Bidoit, Peter D. Mosses, CoFI 81 Casl Tutorial



spec Set to List [ sort Elem ] =

List Set [ sort Elem ]

then op list of : Set → List

∀S : Set • elements of (list of S ) = S

end

➤ Renaming may be used to avoid unintended name clashes, or to adjust names of

sorts and change notations for operations and predicates.

spec Stack [ sort Elem ] =

List Selectors [ sort Elem ] with sort List 7→ Stack ,

ops cons 7→ push onto ,

head 7→ top,

tail 7→ pop

end

➤ When combining specifications, origins of symbols can be indicated.

spec List Set 1 [ sort Elem ] =

List Selectors [ sort Elem ] with empty , cons

and Generated Set [ sort Elem ] with empty , { }, ∪

then op elements of : List → Set

∀e : Elem; L : List

• elements of empty = empty

• elements of cons(e,L) = {e} ∪ elements of L

end

➤ Auxiliary symbols used in structured specifications can be hidden.

spec Natural Partial Subtraction 3 =

Natural Partial Subtraction 1 hide suc, pre

end

spec Natural Partial Subtraction 4 =

Natural Partial Subtraction 1

reveal Nat , 0 , 1 , + , − , ∗ , <

end

spec Partial Order 2 = Partial Order reveal pred ≤

➤ Auxiliary symbols can be made local when they do not need to be exported.

spec List Order [Total Order with sort Elem, pred < ] =

List Selectors [ sort Elem ]

then local op insert : Elem × List → List

∀e, e ′ : Elem; L : List

• insert(e, empty) = cons(e, empty)

• insert(e, cons(e ′,L)) = cons(e ′, insert(e,L)) when e ′ < e

else cons(e, cons(e ′,L))

within op order : List → List

∀e : Elem; L : List

• order(empty) = empty

• order(cons(e,L)) = insert(e, order(L))

end

spec List Order Sorted

[Total Order with sort Elem, pred < ] =

List Selectors [ sort Elem ]

then local pred is sorted : List

∀e, e ′ : Elem; L : List

• empty is sorted

• cons(e, empty) is sorted

• cons(e, cons(e ′,L)) is sorted ⇔

cons(e ′,L) is sorted ∧ ¬(e ′ < e)

within op order : List → List

∀L : List • order(L) is sorted

end

➤ Care is needed with local sort declarations.

spec Wrong List Order Sorted

[Total Order with sort Elem, pred < ] =

List Selectors [ sort Elem ]

then local pred is sorted : List

sort SortedList = {L : List • L is sorted}

∀e, e ′ : Elem; L : List

• empty is sorted

• cons(e, empty) is sorted

• cons(e, cons(e ′,L)) is sorted ⇔

cons(e ′,L) is sorted ∧ ¬(e ′ < e)

within op order : List → SortedList

end

spec List Order Sorted 2

[Total Order with sort Elem, pred < ] =

List Selectors [ sort Elem ]

then local pred is sorted : List

∀e, e ′ : Elem; L : List

• empty is sorted

• cons(e, empty) is sorted

• cons(e, cons(e ′,L)) is sorted ⇔

cons(e ′,L) is sorted ∧ ¬(e ′ < e)

within sort SortedList = {L : List • L is sorted}

op order : List → SortedList

end

spec List Order Sorted 3

[Total Order with sort Elem, pred < ] =

List Selectors [ sort Elem ]

then { pred is sorted : List

∀e, e ′ : Elem; L : List

• empty is sorted

• cons(e, empty) is sorted

• cons(e, cons(e ′,L)) is sorted ⇔

cons(e ′,L) is sorted ∧ ¬(e ′ < e)

then sort SortedList = {L : List • L is sorted}

op order : List → SortedList

} hide is sorted

end

➤ Naming a specification allows its reuse.

It is in general advisable to define as many named specifications as felt appropriate,

since this improves the reusability of specifications: a named specification can easily be

reused by referring to its name.

Generic Specifications

➤ Making a specification generic (when appropriate) improves its reusability.

➤ Parameters are arbitrary specifications.

spec Generic Monoid [ sort Elem ] = . . .

spec List Selectors [ sort Elem ] = . . .

spec List Order [Total Order with sort Elem, pred < ] = . . .

➤ The argument specification of an instantiation must provide symbols corresponding

to those required by the parameter.

spec List Order Nat = List Order [Natural Order ]

➤ The argument specification of an instantiation must ensure that the properties

required by the parameter hold.

spec Nat Word = Generic Monoid [Natural ]

spec List Order Nat = List Order [Natural Order ]

The definition of Nat Word abbreviates:

Natural and { Non Generic Monoid with Elem 7→ Nat }.

➤ When convenient, an instantiation can be completed by a renaming.

spec Nat Word 1 =

Generic Monoid [Natural ]

with Monoid 7→ Nat Word

end

➤ There must be no shared symbols between the argument specification and the body

of the instantiated generic specification.

spec This Is Wrong = Generic Monoid [Monoid ]

The above instantiation is ill-formed since the sort Monoid and the operation symbols

‘1 ’ and ‘∗’ are shared between the body of the generic specification Generic Monoid

and the argument specification Monoid.

➤ In instantiations, the fitting of parameter symbols to identical argument symbols

can be left implicit.

spec Generic Commutative Monoid [ sort Elem ] =

Generic Monoid [ sort Elem ]

then . . .

➤ The fitting of parameter sorts to unique argument sorts can also be left implicit.

➤ Fitting of operation and predicate symbols can sometimes be left implicit too, and

can imply fitting of sorts.

spec List Order Positive = List Order [Positive ]

➤ The intended fitting of the parameter symbols to the argument symbols may have

to be specified explicitly.

spec Nat Word 2 =

Generic Monoid [Natural Subsorts fit Elem 7→ Nat ]

c© Michel Bidoit, Peter D. Mosses, CoFI 99 Casl Tutorial



➤ A generic specification may have more than one parameter.

spec Pair [ sort Elem1 ] [ sort Elem2 ] =

free type Pair ::= pair(first : Elem1 ; second : Elem2 )

spec Table [ sort Key ] [ sort Val ] = . . .

Note that writing:

spec Homogeneous Pair 1 [ sort Elem ] [ sort Elem ] =

free type Pair ::= pair(first : Elem; second : Elem)

merely defines pairs of values of the same sort, and Homogeneous Pair 1 is

(equivalent to and) better defined as follows:

spec Homogeneous Pair [ sort Elem ] =

free type Pair ::= pair(first : Elem; second : Elem)

➤ Instantiation of generic specifications with several parameters is similar to the case

of just one parameter.

spec Pair Natural Color =

Pair [Natural Arithmetic ] [Color fit Elem2 7→ RGB ]

Using the specification Pair 1 (similar to Pair, but with one single parameter

introducing two sorts Elem1 and Elem2 ), would require us to write:

spec Pair Natural Color 1 =

Pair 1 [Natural Arithmetic and Color

fit Elem1 7→ Nat , Elem2 7→ RGB ]

➤ When parameters are trivial, one can always avoid explicit fitting maps.

spec Pair Natural Color 2 =

Pair [ sort Nat ] [ sort RGB ]

and Natural Arithmetic and Color

Compare for instance:

spec Pair Pos =

Homogeneous Pair [ sort Pos] and Integer Arithmetic 1

with:

spec Pair Pos 1 =

Homogeneous Pair [ Integer Arithmetic 1 fit Elem 7→ Pos ]

Note that the instantiation:

Homogeneous Pair 1 [Natural ] [Color fit Elem 7→ RGB ]

is ill-formed, since it entails mapping the sort Elem to both Nat and RGB .

➤ It is easy to specialize a generic specification with several parameters using a

“partial instantiation”.

spec My Table [ sort Val ] =

Table [Natural Arithmetic ] [ sort Val ]

➤ Composition of generic specifications is expressed using instantiation.

spec Set of List [ sort Elem ] =

Generated Set [List Selectors [ sort Elem ] fit Elem 7→ List ]

Note especially that the following specification:

spec Mistake [ sort Elem ] =

Generated Set [List Selectors [ sort Elem ] ]

does not provide sets of lists of elements.

spec Set and List [ sort Elem ] =

Generated Set [ sort Elem ] and List Selectors [ sort Elem ]

It may be worth mentioning that the following composition of generic specifications is

ill-formed:

spec This Is Still Wrong =

Generic Monoid [ Generic Monoid [ sort Elem ]

fit Elem 7→ Monoid ]

➤ Compound sorts introduced by a generic specification get automatically renamed on

instantiation, which avoids name clashes.

spec List Rev [ sort Elem ] =

free type List [Elem] ::= empty |

cons(head :? Elem; tail :? List [Elem])

ops ++ : List [Elem] × List [Elem] → List [Elem],

assoc, unit empty ;

reverse : List [Elem] → List [Elem]

∀e : Elem; L,L1 ,L2 : List [Elem]

• cons(e,L1 ) ++ L2 = cons(e,L1 ++ L2 )

• reverse(empty) = empty

• reverse(cons(e,L)) = reverse(L) ++ cons(e, empty)

end

spec List Rev Nat = List Rev [Natural ]

spec Two Lists =

List Rev [Natural ] %% Provides the sort List [Nat ]

and List Rev [Color fit Elem 7→ RGB ] %% Provides the sort List [RGB ]

spec Two Lists 1 =

List Rev [ Integer Arithmetic 1 fit Elem 7→ Nat ]

and List Rev [ Integer Arithmetic 1 fit Elem 7→ Int ]

Remember that Nat < Int does not entail List [Nat ] < List [Int ].

spec Monoid C [ sort Elem ] =

sort Monoid [Elem]

ops inj : Elem → Monoid [Elem];

1 : Monoid [Elem];

∗ : Monoid [Elem] × Monoid [Elem] → Monoid [Elem],

assoc, unit 1

∀x , y : Elem • inj (x ) = inj (y) ⇒ x = y

end

spec Monoid of Monoid [ sort Elem ] =

Monoid C [Monoid C [ sort Elem ] fit Elem 7→ Monoid [Elem] ]

➤ Compound symbols can also be used for operations and predicates.

spec List Rev Order [Total Order ] =

List Rev [ sort Elem ]

then local op insert : Elem × List [Elem] → List [Elem]

∀e, e ′ : Elem; L : List [Elem]

• insert(e, empty) = cons(e, empty)

• insert(e, cons(e ′,L)) = cons(e ′, insert(e,L)) when e ′ < e

else cons(e, cons(e ′,L))

within op order [ < ] : List [Elem] → List [Elem]

∀e : Elem; L : List [Elem]

• order [ < ](empty) = empty

• order [ < ](cons(e,L)) = insert(e, order [ < ](L))

end

spec List Rev with Two Orders =

List Rev Order

[ Integer Arithmetic Order fit Elem 7→ Int , < 7→ < ]

%% Provides the sort List [Int ] and the operation order [ < ]

and List Rev Order

[ Integer Arithmetic Order fit Elem 7→ Int , < 7→ > ]

%% Provides the sort List [Int ] and the operation order [ > ]

then %implies

∀L : List [Int ] • order [ < ](L) = reverse(order [ > ](L))

end

➤ Parameters should be distinguished from references to fixed specifications that are

not intended to be instantiated.

spec List Weighted Elem [ sort Elem op weight : Elem → Nat ]

given Natural Arithmetic =

List Rev [ sort Elem ]

then op weight : List [Elem] → Nat

∀e : Elem; L : List [Elem]

• weight(empty) = 0

• weight(cons(e,L)) = weight(e) + weight(L)

end

One could have written instead:

spec List Weighted Elem

[Natural Arithmetic then sort Elem op weight : Elem → Nat ] = . . .

but the latter, which is correct, misses the essential distinction between the part which

is intended to be specialized and the part which is ‘fixed’ (since, by definition, the

parameter is the part which has to be specialized).

Note also that omitting the ‘given Natural Arithmetic’ clause would make the

declaration:

spec List Weighted Elem [ sort Elem op weight : Elem → Nat ] = . . .

ill-formed, since the sort Nat is not available.

➤ Argument specifications are always implicitly regarded as extension of the imports.

spec List Weighted Pair Natural Color =

List Weighted Elem [Pair Natural Color fit Elem 7→ Pair ,

weight 7→ first ]

spec List Weighted Instantiated =

List Weighted Elem [ sort Value op weight : Value → Nat ]

➤ Imports are also useful to prevent ill-formed instantiations.

spec List Length [ sort Elem ] given Natural Arithmetic =

List Rev [ sort Elem ]

then op length : List [Elem] → Nat

∀e : Elem; L : List [Elem]

• length(empty) = 0

• length(cons(e,L)) = length(L) + 1

then %implies

∀L : List [Elem] • length(reverse(L)) = length(L)

end

spec List Length Natural =

List Length [Natural Arithmetic ]

spec Wrong List Length [ sort Elem ] =

Natural Arithmetic and List Rev [ sort Elem ]

then . . .

end

The specification Wrong List Length is fine as long as one does not need to

instantiate it with Natural Arithmetic as argument specification.

The instantiation Wrong List Length [Natural Arithmetic ] is ill-formed since

some symbols of the argument specification are shared with some symbols of the body

(and not already occurring in the parameter) of the instantiated generic specification,

which is wrong. Of course the same problem will occur with any argument specification

which provides, e.g., the sort Nat .

➤ In generic specifications, auxiliary required specifications should be imported rather

than extended.

Since an instantiation is ill-formed as soon as there are some shared symbols between

the argument specification and the body of the generic specification, when designing a

generic specification, it is generally advisable to turn auxiliary required specifications

into imports, and generic specifications of the form:

F [X ] = SP then . . .

are better written

F [X ] given SP = . . .

to allow the instantiation F [SP ].

➤ Views are named fitting maps, and can be defined along with specifications.

view Integer as Total Order :

Total Order to Integer Arithmetic Order =

Elem 7→ Int , < 7→ <

view Integer as Reverse Total Order :

Total Order to Integer Arithmetic Order =

Elem 7→ Int , < 7→ >

spec List Rev with Two Orders 1 =

List Rev Order [ view Integer as Total Order ]

and List Rev Order [ view Integer as Reverse Total Order ]

then %implies

∀L : List [Int ] • order [ < ](L) = reverse(order [ > ](L))

end

➤ Views can also be generic.

view List as Monoid [ sort Elem ] :

Monoid to List Rev [ sort Elem ] =

Monoid 7→ List [Elem], 1 7→ empty , ∗ 7→ + +

Specifying the Architecture of

Implementations

➤ Architectural specifications impose structure on implementations, whereas

specification-building operations only structure the text of specifications.

➤ The examples in this chapter are artificially simple.

spec Color = . . .

spec Natural Order = . . .

spec Natural Arithmetic = . . .

spec Elem = sort Elem

spec Cont [Elem ] =

generated type Cont [Elem] ::= empty | insert(Elem; Cont [Elem])

preds is empty : Cont [Elem];

is in : Elem × Cont [Elem]

ops choose : Cont [Elem] →? Elem;

delete : Elem × Cont [Elem] → Cont [Elem]

∀e, e′ : Elem; C : Cont [Elem]

• empty is empty

• ¬ insert(e,C ) is empty

• ¬ e is in empty

• e is in insert(e′,C ) ⇔ (e = e′ ∨ e is in C )

• def choose(C ) ⇔ ¬ C is empty

• def choose(C ) ⇒ choose(C ) is in C

• e is in delete(e′,C ) ⇔ (e is in C ∧ ¬(e = e′))

end

spec Cont Diff [Elem ] =

Cont [Elem ]

then op diff : Cont [Elem] × Cont [Elem] → Cont [Elem]

∀e : Elem; C ,C ′ : Cont [Elem]

• e is in diff (C ,C ′) ⇔ (e is in C ∧ ¬(e is in C ′))

end

spec Req = Cont Diff [Natural Order ]

spec Flat Req =

free type Nat ::= 0 | suc(Nat)

pred < : Nat × Nat

generated type Cont[Nat] ::= empty | insert(Nat; Cont[Nat])

preds is empty : Cont[Nat];

is in : Nat × Cont[Nat]

ops choose : Cont[Nat] →? Nat;

delete : Nat × Cont[Nat] → Cont[Nat];

diff : Cont[Nat] × Cont[Nat] → Cont[Nat]

∀e, e′ : Nat; C , C ′ : Cont[Nat]

• 0 < suc(e)

• ¬(e < 0)

• suc(e) < suc(e′) ⇔ e < e′

• empty is empty

• ¬ insert(e, C ) is empty

• ¬ e is in empty

• e is in insert(e′
, C ) ⇔ (e = e′ ∨ e is in C )

• def choose(C ) ⇔ ¬ C is empty

• def choose(C ) ⇒ choose(C ) is in C

• e is in delete(e′
, C ) ⇔ (e is in C ∧ ¬(e = e′))

• e is in diff (C , C ′) ⇔ (e is in C ∧ ¬(e is in C ′))

end

➤ An architectural specification consists of a list of unit declarations, specifying the

required components, and a result part, indicating how they are to be combined.

arch spec System =

units N : Natural Order;

C : Cont [Natural Order ] given N ;

D : Cont Diff [Natural Order ] given C

result D

➤ There can be several distinct architectural choices for the same requirements

specification.

arch spec System 1 =

units N : Natural Order;

CD : Cont Diff [Natural Order ] given N

result CD

c© Michel Bidoit, Peter D. Mosses, CoFI 124 Casl Tutorial



➤ Each unit declaration listed in an architectural specification corresponds to a

separate implementation task.

In the architectural specification System, the task of providing a component D

expanding C and implementing Cont Diff [Natural Order ] is independent

from the tasks of providing implementations N of Natural Order and

C of Cont [Natural Order ] given N .

Hence, when providing the component D , one cannot make any further assumption on

how the component C is (or will be) implemented, besides what is expressly ensured by

its specification.

Thus the component D should expand any given implementation C of

Cont [Natural Order ] and provide an implementation of

Cont Diff [Natural Order ], which is tantamount to providing a generic

implementation G of Cont Diff [Natural Order ] which takes the particular

implementation of Cont [Natural Order ] as a parameter to be expanded.

Then we obtain D by simply applying G to C .

Genericity here arises from the independence of the developments

of C and D , rather than from the desire to build multiple implementations

of Cont Diff [Natural Order ] using different implementations of

Cont [Natural Order ].

➤ A unit can be implemented only if its specification is a conservative extension of the

specifications of its given units.

For instance, the component D can exist only if the specification

Cont Diff [Natural Order ] is a conservative extension of

Cont [Natural Order ].

spec Cont Diff 1 =

Cont [Natural Order ]

then op diff : Cont [Nat ] × Cont [Nat ] → Cont [Nat ]

∀x , y : Nat ; C ,C ′ : Cont [Nat ]

• diff (C , empty) = C

• diff (empty,C ′) = empty

• diff (insert(x , C ), insert(y ,C ′)) =

insert(x , diff (C , insert(y ,C ′))) when x < y

else diff (C ,C ′) when x = y

else diff (insert(x , C ),C ′)

• x is in diff (C ,C ′) ⇔ (x is in C ∧ ¬(x is in C ′))

end

arch spec Inconsistent =

units N : Natural Order;

C : Cont [Natural Order ] given N ;

D : Cont Diff 1 given C

result D

➤ Genericity of components can be made explicit in architectural specifications.

arch spec System G =

units N : Natural Order;

F : Natural Order → Cont [Natural Order ];

G : Cont [Natural Order ] → Cont Diff [Natural Order ]

result G [F [N ]]

➤ A generic component may be applied to an argument richer than required by its

specification.

arch spec System A =

units NA : Natural Arithmetic;

F : Natural Order → Cont [Natural Order ];

G : Cont [Natural Order ] → Cont Diff [Natural Order ]

result G [F [NA]]

➤ Specifications of components can be named for further reuse.

unit spec Cont Comp = Elem → Cont [Elem ]

unit spec Diff Comp = Cont [Elem ] → Cont Diff [Elem ]

arch spec System G1 =

units N : Natural Order;

F : Cont Comp;

G : Diff Comp

result G [F [N ]]

➤ Both named and un-named specifications can be used to specify components.

unit spec Diff Comp 1 =

Cont [Elem ] → { op diff : Cont [Elem] × Cont [Elem] → Cont [Elem]

∀e : Elem; C ,C ′ : Cont [Elem]

• e is in diff (C ,C ′) ⇔

(e is in C ∧ ¬(e is in C ′)) }

➤ Specifications of generic components should not be confused with generic

specifications.

• Generic specifications naturally give rise to specifications of generic components,

which can be named for later reuse, as illustrated above by Cont Comp.

• A generic specification is nothing other than a piece of specification that can

easily be adapted by instantiation.

• A specification of a generic component cannot be instantiated,

it is the specified generic component which gets applied to suitable components.

➤ A generic component may be applied more than once in the same architectural

specification.

arch spec Other System =

units N : Natural Order;

C : Color;

F : Cont Comp

result F [N ] and F [C fit Elem 7→ RGB ]

➤ Several applications of the same generic component is different from applications of

several generic components with similar specifications.

arch spec Other System 1 =

units N : Natural Order;

C : Color;

FN : Natural Order → Cont [Natural Order ];

FC : Color → Cont [Color fit Elem 7→ RGB ]

result FN [N ] and FC [C ]

➤ Generic components may have more than one argument.

unit spec Set Comp = Elem → Generated Set [Elem ]

spec Cont2Set [Elem ] =

Cont [Elem ] and Generated Set [Elem ]

then op elements of : Cont [Elem] → Set

∀e : Elem; C : Cont [Elem]

• elements of empty = empty

• elements of insert(e,C ) = {e} ∪ elements of C

end

arch spec Arch Cont2Set Nat =

units N : Natural Order;

C : Cont Comp;

S : Set Comp;

F : Cont [Elem ] × Generated Set [Elem ] → Cont2Set [Elem ]

result F [C [N ]] [S [N ]]

➤ Open systems can be described by architectural specifications using generic unit

expressions in the result part.

arch spec Arch Cont2Set =

units C : Cont Comp;

S : Set Comp;

F : Cont [Elem ] × Generated Set [Elem ] → Cont2Set [Elem ]

result λX : Elem • F [C [X ]] [S [X ]]

arch spec Arch Cont2Set Used =

units N : Natural Order;

CSF : arch spec Arch Cont2Set

result CSF [N ]

➤ When components are to be combined, it is best to check that any shared symbol

originates from the same non-generic component.

arch spec Arch Cont2Set Nat 1 =

units N : Natural Order;

C : Cont Comp;

S : Set Comp;

G : { Cont [Elem ] and Generated Set [Elem ] }

→ Cont2Set [Elem ]

result G [C [N ] and S [N ] fit Cont [Elem] 7→ Cont [Nat ] ]

arch spec Wrong Arch Spec =

units CN : Cont [Natural Order ];

SN : Generated Set [Natural Order ];

F : Cont [Elem ] × Generated Set [Elem ] → Cont2Set [Elem ]

result F [CN ] [SN ]

arch spec Badly Structured Arch Spec =

units N : Natural Order;

A : Natural Order → Natural Arithmetic;

C : Cont Comp;

S : Set Comp;

F : Cont [Elem ] × Generated Set [Elem ] → Cont2Set [Elem ]

result F [C [A [N ]]] [S [A [N ]]]

➤ Auxiliary unit definitions or local unit definitions may be used to avoid repetition of

generic unit applications.

arch spec Well Structured Arch Spec =

units N : Natural Order;

A : Natural Order → Natural Arithmetic;

AN = A [N ];

C : Cont Comp;

S : Set Comp;

F : Cont [Elem ] × Generated Set [Elem ] → Cont2Set [Elem ]

result F [C [AN ]] [S [AN ]]

arch spec Another Well Structured Arch Spec =

units N : Natural Order;

A : Natural Order → Natural Arithmetic;

C : Cont Comp;

S : Set Comp;

F : Cont [Elem ] × Generated Set [Elem ] → Cont2Set [Elem ]

result local AN = A [N ] within F [C [AN ]] [S [AN ]]

Libraries

➤ Libraries are named collections of named specifications.

c© Michel Bidoit, Peter D. Mosses, CoFI 142 Casl Tutorial



➤ Local libraries are self-contained.

A library is called local when it is self-contained, i.e., for each reference to a

specification name in the library, the library includes a specification with that name.

➤ Distributed libraries support reuse.

Distributed libraries allow duplication of specifications to be avoided altogether.

Instead of making an explicit copy of a named specification from one library for use in

another, the second library merely indicates that the specification concerned can be

downloaded from the first one.

➤ Different versions of the same library are distinguished by hierarchical version

numbers.

➤ Local libraries are self-contained collections of specifications.

library UserManual/Examples

. . .

spec Natural = . . .

. . .

spec Natural Order = Natural then . . .

. . .

➤ Specifications can refer to previous items in the same library.

library UserManual/Examples

. . .

spec Strict Partial Order = . . .

. . .

spec Total Order = Strict Partial Order then . . .

. . .

spec Partial Order = Strict Partial Order then . . .

. . .

➤ All kinds of named specifications can be included in libraries.

library UserManual/Examples

. . .

spec Strict Partial Order = . . .

. . .

spec Generic Monoid [ sort Elem ] = . . .

. . .

view Integer as Total Order : . . .

. . .

view List as Monoid [ sort Elem ] : . . .

. . .

arch spec System = . . .

. . .

unit spec Cont Comp = . . .

. . .

➤ Display, parsing, and literal syntax annotations apply to entire libraries.

library UserManual/Examples

. . .

%display <= %LATEX ≤

%display >= %LATEX ≥

%display union %LATEX ∪

%prec { + , − } < { ∗ }

%left assoc + , ∗

. . .

spec Strict Partial Order = . . .

. . .

spec Partial Order = Strict Partial Order then . . .≤. . .

. . .

spec Generated Set [ sort Elem ] = . . .∪. . .

. . .

spec Integer Arithmetic Order = . . .≤. . .≥. . .

. . .

Parsing annotations allow omission of grouping parentheses when terms are input. A

single annotation can indicate the relative precedence or the associativity (left or right)

of a group of operation symbols. The precedence annotation for infix arithmetic

operations given above, namely:

%prec { + , − } < { ∗ }

allows a term such as a + (b ∗ c) to be input (and hence also displayed) as a + b ∗ c.

The left-associativity annotation for + and ∗:

%left assoc + , ∗

allows (a + b) + c to be input as a + b + c, and similarly for ∗; but the parentheses

cannot be omitted in (a + b) − c (not even if ‘ − ’ were to be included in the same

left-associativity annotation).

When an operation symbol is declared with the associativity attribute assoc, an

associativity annotation for that symbol is provided automatically.

➤ Libraries and library items can have author and date annotations.

library UserManual/Examples

%authors( Michel Bidoit 〈bidoit@lsv.ens-cachan.fr〉,

Peter D. Mosses 〈pdmosses@brics.dk〉 )%

%dates 15 Oct 2003, 1 Apr 2000

. . .

spec Strict Partial Order = . . .

. . .

%authors Michel Bidoit 〈bidoit@lsv.ens-cachan.fr〉

%dates 10 July 2003

spec Integer Arithmetic Order =

. . .

c© Michel Bidoit, Peter D. Mosses, CoFI 150 Casl Tutorial

<bidoit@lsv.ens-cachan.fr>
<pdmosses@brics.dk>
<bidoit@lsv.ens-cachan.fr>


➤ Libraries can be installed on the Internet for remote access. Validated libraries can

be registered for public access.

library Basic/Numbers

. . .

%left assoc @@

%number @@

%floating ::: , E

%prec { E } < { ::: }

. . .

spec Nat =

free type Nat ::= 0 | suc(Nat)

. . .

ops 1 : Nat = suc(0 ); . . . ; 9 : Nat = suc(8 );

@@ (m,n : Nat) : Nat = (m ∗ suc(9 )) + n
. . .

c© Michel Bidoit, Peter D. Mosses, CoFI 151 Casl Tutorial



spec Int = Nat then . . .

spec Rat = Int then . . .

spec DecimalFraction = Rat then

. . .

ops ::: : Nat × Nat → Rat ;

E : Rat × Int → Rat
. . .

➤ Libraries should include appropriate annotations.

➤ Libraries can include items downloaded from other libraries.

library Basic/StructuredDatatypes

. . .

from Basic/Numbers get Nat, Int

. . .

spec List [ sort Elem ] given Nat = . . .

. . .

spec Array . . . given Int = . . .

. . .

from Basic/Numbers get Nat 7→ Natural, Int 7→ Integer

➤ Substantial libraries of basic datatypes are already available.

Basic/Numbers: natural numbers, integers, and rationals.

Basic/RelationsAndOrders: reflexive, symmetric, and transitive relations,

equivalence relations, partial and total orders, boolean algebras.

Basic/Algebra I: monoids, groups, rings, integral domains, and fields.

Basic/SimpleDatatypes: booleans, characters.

Basic/StructuredDatatypes: sets, lists, strings, maps, bags, arrays, trees.

Basic/Graphs: directed graphs, paths, reachability, connectedness, colorability, and

planarity.

Basic/Algebra II: monoid and group actions on a space, euclidean and factorial

rings, polynomials, free monoids, and free commutative monoids.

Basic/LinearAlgebra I: vector spaces, bases, and matrices.

Basic/LinearAlgebra II: algebras over a field.

Basic/MachineNumbers: bounded subtypes of naturals and integers.

➤ Libraries need not be registered for public access.

library http://www.cofi.info/CASL/Test/Security

. . .

from http://casl:password@www.cofi.info/CASL/RSA get Key

. . .

spec Decrypt = Key then . . .

. . .

➤ Subsequent versions of a library are distinguished by explicit version numbers.

library Basic/Numbers version 1.0

. . .

spec Nat = . . .

. . .

spec Int = Nat then . . .

. . .

spec Rat = Int then . . .

. . .

➤ Libraries can refer to specific versions of other libraries.

library Basic/StructuredDatatypes version 1.0

. . .

from Basic/Numbers version 1.0 get Nat, Int

. . .

spec List [ sort Elem ] given Nat = . . .

. . .

spec Array . . . given Int = . . .

. . .

➤ All downloadings should be collected at the beginning of a library.

Tools

➤ The Heterogeneous Tool Set (HETS) is the main analysis tool for CASL.

• CASL specifications can also be checked for well-formedness using a form-based web

page.

• HETS can be used for parsing and checking static well-formedness of specifications.

• HETS also displays and manages proof obligations, using development graphs.

• Nodes in a development graph correspond to CASL specifications.

Arrows show how specifications are related by the structuring constructs.

• Internal nodes in a development graph correspond to unnamed parts of a structured

specification.

• HOL-CASL is an interactive theorem prover for CASL, based on the tactical theorem

prover ISABELLE.

• CASL is linked to ISABELLE/HOL by an encoding.

• ASF+SDF was used to prototype the CASL syntax.

• The ASF+SDF Meta-Environment provides syntax-directed editing of CASL

specifications.

